
Listas

Construcción de listas

Construcción de listas
I Definición de listas:

I La lista vacía [] es una lista.
I Si L es una lista, entonces .(a,L) es una lista.

I Ejemplos:
?- .(X,Y) = [a].

X = a

Y = []

?- .(X,Y) = [a,b].

X = a

Y = [b]

?- .(X,.(Y,Z)) = [a,b].

X = a

Y = b

Z = []

Listas

Construcción de listas

Escritura abreviada
I Escritura abreviada:

[X|Y] = .(X,Y)

I Ejemplos con escritura abreviada:
?- [X|Y] = [a,b].

X = a

Y = [b]

?- [X|Y] = [a,b,c,d].

X = a

Y = [b, c, d]

?- [X,Y|Z] = [a,b,c,d].

X = a

Y = b

Z = [c, d]

Listas

Definición de relaciones sobre listas

Definición de concatenación (append)
I Especificación: conc(A,B,C) se verifica si C es la lista obtenida

escribiendo los elementos de la lista B a continuación de los
elementos de la lista A. Por ejemplo,
?- conc([a,b],[b,d],C).

C =[a,b,b,d]

I Definición 1:

conc(A,B,C) :- A=[], C=B.

conc(A,B,C) :- A=[X|D], conc(D,B,E), C=[X|E].

I Definición 2:

conc([],B,B).

conc([X|D],B,[X|E]) :- conc(D,B,E).

Definición de relaciones sobre listas

Consultas con la relación de concatenación
I Analogía entre la definición de conc y la de suma,
I ¿Cuál es el resultado de concatenar las listas [a,b] y [c,d,e]?

?- conc([a,b],[c,d,e],L).

L = [a, b, c, d, e]
I ¿Qué lista hay que añadirle a la lista [a,b] para obtener

[a,b,c,d]?
?- conc([a,b],L,[a,b,c,d]).

L = [c, d]
I ¿Qué dos listas hay que concatenar para obtener [a,b]?

?- conc(L,M,[a,b]).

L = [] M = [a, b] ;

L = [a] M = [b] ;

L = [a, b] M = [] ;

No

Listas

Definición de relaciones sobre listas

Árbol de deducción de ?- conc(L,M,[a,b]).

Definición de relaciones sobre listas

Definición de la relación de pertenencia (member)
I Especificación: pertenece(X,L) se verifica si X es un elemento

de la lista L.
I Definición 1:

pertenece(X,[X|L]).

pertenece(X,[Y|L]) :- pertenece(X,L).

I Definición 2:

pertenece(X,[X|_]).

pertenece(X,[_|L]) :- pertenece(X,L).

Listas

Definición de relaciones sobre listas

Consultas con la relación de pertenencia

?- pertenece(b,[a,b,c]).

Yes

?- pertenece(d,[a,b,c]).

No

?- pertenece(X,[a,b,a]).

X = a ;

X = b ;

X = a ;

No

?- pertenece(a,L).

L = [a|_G233] ;

L = [_G232, a|_G236] ;

L = [_G232, _G235, a|_G239]

Yes

Listas

Definición de relaciones sobre listas

Árbol de deducción de ?- pertenece(a,L).

Disyunciones

Disyunciones
I Definición de pertenece con disyunción

pertenece(X,[Y|L]) :- X=Y ; pertenece(X,L).

I Definición equivalente sin disyunción

pertenece(X,[Y|L]) :- X=Y.

pertenece(X,[Y|L]) :- pertenece(X,L).

Ejemplos de operadores aritméticos
I Ejemplos de notación infija y prefija en expresiones aritméticas:

?- +(X,Y) = a+b.

X = a

Y = b

?- +(X,Y) = a+b+c.

X = a+b

Y = c

?- +(X,Y) = a+(b+c).

X = a

Y = b+c

?- a+b+c = (a+b)+c.

Yes

?- a+b+c = a+(b+c).

No

Operadores aritméticos

Ejemplos de asociatividad y precedencia
I Ejemplos de asociatividad:

?- X^Y = a^b^c.

X = a Y = b^c

?- a^b^c = a^(b^c).

Yes

I Ejemplo de precedencia
?- X+Y = a+b*c.

X = a Y = b*c

?- X*Y = a+b*c.

No

?- X*Y = (a+b)*c.

X = a+b Y = c

?- a+b*c = (a+b)*c.

No

Operadores aritméticos

Operadores aritméticos predefinidos

Precedencia Tipo Operadores
500 yfx +,- Infijo asocia por la izquierda
500 fx - Prefijo no asocia
400 yfx *, / Infijo asocia por la izquierda
200 xfy ^ Infijo asocia por la derecha

Definición de operadores

Definición de operadores
I Definición (ejemplo_operadores.pl)

:-op(800,xfx,estudian).

:-op(400,xfx,y).

juan y ana estudian lógica.

I Consultas
?- [ejemplo_operadores].

?- Quienes estudian lógica.

Quienes = juan y ana

?- juan y Otro estudian Algo.

Otro = ana

Algo = lógica

Evaluación de expresiones aritméticas

Evaluación de expresiones aritméticas
I Evaluación de expresiones aritmética con is.

?- X is 2+3^3.

X = 29

?- X is 2+3, Y is 2*X.

X = 5

Y = 10
I Relaciones aritméticas: <, =<, >, >=, =:= y =/=

?- 3 =< 5.

Yes

?- 3 > X.

% [WARNING: Arguments are not sufficiently instantiated]

?- 2+5 = 10-3.

No

?- 2+5 =:= 10-3.

Yes

Definición de relaciones aritméticas

Definición del factorial
I factorial(X,Y) se verifica si Y es el factorial de X. Por ejemplo,

?- factorial(3,Y).

Y = 6 ;

No

I Definición:

factorial(1,1).

factorial(X,Y) :-

X > 1,

A is X - 1,

factorial(A,B),

Y is X * B.

Aritmética

Definición de relaciones aritméticas

Árbol de deducción de ?- factorial(3,Y).

